Series

Los objetivos de esta práctica son:

• Estudiar el carácter de series.

• Hallar la suma de algunas series.

1. Comentarios

Criterio logarítmico: un criterio útil para estudiar el carácter de una serie con Maple

Los criterios de comparación de series tienen el inconveniente de que no son intrínsecos. Es decir, es necesario encontrar otra serie con la que comparar. Por el contrario, otros criterios como el del cociente (o de D'Alembert) y el de la raíz (o de Cauchy) no dependen más que de la serie que estudiamos. Sin embargo, a veces el límite que hay que calcular para aplicar estos criterios es más complicado. Pero cuando el límite lo calculamos con Maple (si nos fiamos), esta complicación no nos importa. Por eso, citamos aquí otro criterio (un **criterio logarítmico**) que puede ser útil cuando el cálculo del límite se lo dejamos a Maple.

Sea $\sum_{n=1}^{\infty} a(n)$ una serie y sea $r = \lim_{n\to\infty}\left(-\frac{\log|a(n)|}{\log n}\right)$, si el límite existe (real, $+\infty$, $-\infty$).

Entonces:

• Si $1 < r$, la serie $\sum_{n=1}^{\infty}|a(n)|$ converge.

• Si $r < 1$, la serie $\sum_{n=1}^{\infty}|a(n)|$ no converge.

• Si $r < 0$, la serie $\sum_{n=1}^{\infty} a(n)$ no converge (el término general no tiende a 0).

Las principales **ventajas** de este criterio son:

• Si el criterio de D'Alembert o el de Cauchy dan resultado, entonces este criterio también lo da. En concreto:

Si $\lim_{n\to\infty}\left|\frac{a(n+1)}{a(n)}\right| < 1$, entonces $\lim_{n\to\infty}\left(-\frac{\log|a(n)|}{\log n}\right) = \infty$ (y lo análogo con el de Cauchy).

Si $1 < \lim_{n\to\infty}\left|\frac{a(n+1)}{a(n)}\right|$, entonces $\lim_{n\to\infty}\left(-\frac{\log|a(n)|}{\log n}\right) = -\infty$ (y lo análogo con el de Cauchy).

• Si $0 < \lim_{n\to\infty}\frac{|a(n)|}{n^{-r}} < \infty$, entonces $\lim_{n\to\infty}\left(-\frac{\log|a(n)|}{\log n}\right) = r$.

El principal **inconveniente** de este criterio es que puede ser difícil de calcular.

Uso de la orden `simplify`

Los criterios más útiles para estudiar el carácter de una serie tienen que ver con límites y es muy frecuente que para calcularlos haya que simplificar antes la expresión. Por ejemplo, veamos el carácter de la serie $\sum_{n=1}^{\infty} n^2x^n$. Intentamos aplicar el criterio del cociente (o de D'Alembert):

```
> restart:
> a:=n->n^2*x^n;
```

$$a := n \to n^2\, x^n$$

```
> limit(abs(a(n+1)/a(n)),n=infinity);
```

$$\lim_{n\to\infty}\left|\frac{(n+1)^2\, x^{n+1}}{n^2\, x^n}\right|$$

Maple no sabe calcular este límite tan sencillo. El problema parece que consiste en que no simplifica, porque si le obligamos a simplificar, entonces sí halla el límite:

```
> simplify(abs(a(n+1)/a(n)));
```

$$\left|\frac{(n+1)^2\, x}{n^2}\right|$$

```
> limit(simplify(abs(a(n+1)/a(n))),n=infinity);
```

$$|x|$$

Podemos presentarlo de manera más legible usando la orden `Limit` (observemos que solo es necesario escribir `simplify` en la orden `limit`, que es la que hace el cálculo):

```
> Limit(abs(a(n+1)/a(n)),n=infinity)=
  limit(simplify(abs(a(n+1)/a(n))),n=infinity);
```

$$\lim_{n\to\infty}\left|\frac{(n+1)^2\, x^{n+1}}{n^2\, x^n}\right| = |x|$$

Del criterio de D'Alembert, deducimos que converge si $|x| < 1$, y que no converge si $1 < |x|$. Por último, la serie tampoco converge si $|x| = 1$, porque el término general n^2x^n no tiende a 0.

2. Carácter de una serie

Ejercicios resueltos

1) Estudiar el carácter de la serie $\sum_{n=1}^{\infty}\frac{1}{\sqrt{n}-\frac{2}{3}}$.

Esta serie es de términos positivos.

```
> restart;
> Limit((1/(sqrt(n)-2/3))/(1/sqrt(n)),n=infinity)=
  limit((1/(sqrt(n)-2/3))/(1/sqrt(n)),n=infinity);
```

$$\lim_{n\to\infty}\frac{\sqrt{n}}{\sqrt{n}-\frac{2}{3}} = 1$$

Como la serie $\sum_{n=1}^{\infty}\frac{1}{\sqrt{n}}$ diverge, se deduce que la serie $\sum_{n=1}^{\infty}\frac{1}{\sqrt{n}-\frac{2}{3}}$ también diverge.

Otra forma de resolver el problema es mediante el criterio logarítmico:

```
> r:=-limit(log(1/(sqrt(n)-2/3))/log(n),n=infinity);
```

$$r := \frac{1}{2}$$

Luego la serie diverge ($r < 1$).

2) Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \frac{1+n^2}{n!}$.

Vamos a usar el criterio del cociente (o de D'Alembert):

```
> restart;a:=n->(1+n^2)/n!:
> Limit(abs(a(n+1)/a(n)),n=infinity)=
  limit(abs(a(n+1)/a(n)),n=infinity);
```

$$\lim_{n\to\infty} \left| \frac{\left(1+(n+1)^2\right) n!}{(n+1)!\,\left(1+n^2\right)} \right| = 0$$

Como el límite es menor que 1, la serie $\sum_{n=1}^{\infty} \frac{1+n^2}{n!}$ converge.

También se puede probar mediante el criterio logarítmico:

```
> r:=-limit(log(a(n))/log(n),n=infinity);
```

$$r := \infty$$

Luego, en efecto, la serie converge ($1 < r$).

3) Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \left(\frac{n+1}{n}\right)^{-n^3}$.

```
> restart;a:=n->((n+1)/n)^(-n^3):
```

Usamos el criterio logarítmico:

```
> r:=-limit(log(a(n))/log(n),n=infinity);
```

$$r := \infty$$

Así que la serie converge ($1 < r$).

4) Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \frac{sen(n)^4}{n^2}$.

Aplicamos el criterio logarítmico:

```
> restart;a:=n->sin(n)**4/n**2;
  r:=limit(-log(abs(a(n)))/log(n),n=infinity);
```

$$a := n \to \frac{\sin(n)^4}{n^2}$$

$$r := \mathrm{O}(1)$$

Maple no sabe calcular este límite. ¿Cuál es el problema? ¿No existe el límite? Realmente, podríamos probar que no existe; una explicación es que hay valores de la función $sen(n)$ muy próximos a 0, y para ellos $-\frac{\log|a(n)|}{\log n}$ es muy grande, mientras que también hay valores de la función $sen(n)$ muy próximos a 1 y para ellos $-\frac{\log|a(n)|}{\log n}$ es próximo a 2. De todas formas, como el criterio logarítmico no da resultado intentamos otro método. Es inmediato que

$$\left|\frac{sen(n)^4}{n^2}\right| \le \frac{1}{n^2}$$

y como $\sum_{n=1}^{\infty} \frac{1}{n^2}$ converge, deducimos que $\sum_{n=1}^{\infty} \frac{sen(n)^4}{n^2}$ también converge (absolutamente).

3. Suma de una serie

5) Calcular la suma de la serie $\sum_{n=1}^{\infty} \frac{1}{n(n+1)}$, si es que converge.

La serie converge, porque $0 \le \frac{1}{n(n+1)}$, $\frac{1}{n(n+1)} \le \frac{1}{n^2}$ y la serie $\sum_{n=1}^{\infty} \frac{1}{n^2}$ converge.

En cuanto al valor de la suma, la orden es muy sencilla:

```
> sum(1/n/(n+1),n=1..infinity);
```

$$1$$

La orden **Sum** devuelve la serie, pero sin calcular su valor:

```
> Sum(1/n/(n+1),n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n(n+1)}$$

Podemos aprovechar ambas órdenes, para que quede mejor indicado:

```
> Sum(1/n/(n+1),n=1..infinity)=sum(1/n/(n+1),n=1..infinity);
```

$$\sum_{n=1}^{\infty} \frac{1}{n(n+1)} = 1$$

Sin embargo, Maple solo sabe calcular la suma de algunos tipos de series. Puede ocurrir que no sepa hallar una suma, o que los cálculos lleven mucho tiempo, o incluso (esto es lo peor) que el programa se quede "colgado". Por esta razón, hay que ser especialmente prudente y guardar el documento (File/Save, o Ctrl+S) antes de introducir la orden **sum**.